unittest
{
	assert( [ "5", "1 3 3 3 3", "3 3 2 2 2" ].parse.expand.solve == 4 );
	assert( [ "5", "1 1 1 2 2", "3 2 1 1 1" ].parse.expand.solve == 0 );
	assert( [ "10", "1 3776 3776 8848 8848 8848 8848 8848 8848 8848", "8848 8848 8848 8848 8848 8848 8848 8848 3776 5" ].parse.expand.solve == 884111967 );
	assert( [ "1", "17", "17" ].parse.expand.solve == 1 );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	input.popFront;
	auto ts = input.front.split.to!( long[] );
	input.popFront;
	auto as = input.front.split.to!( long[] );
	return tuple( ts, as );
}

auto solve( long[] ts, long[] as )
{
	const MOD = 10L ^^ 9 + 7;
	
	// 高さが確定する山を判定
	auto ms = new long[ ts.length ];
	auto p = 0L;
	foreach( i, ref m; ms )
	{
		if( p < ts[ i ] )
		{
			if( as[ i ] < ts[ i ] ) return 0;		// 矛盾するのでダメ
			m = 1;		// 確定したときは1(パターン数)を入れておく
		}
		else
		{
			m = ts[ i ];		// 確定しないときはtとaの低いほうを入れておく
		}
		p = ts[ i ];
	}
	p = 0L;
	foreach_reverse( i, ref m; ms )
	{
		if( p < as[ i ] )
		{
			if( ts[ i ] < as[ i ] ) return 0;		// 矛盾するのでダメ
			m = 1;		// 確定したときは1(パターン数)を入れておく
		}
		else
		{
			m = min( m, as[ i ] );		// 確定しないときはtとaの低いほうを入れておく
		}
		p = as[ i ];
	}
	
	// 山脈のパターン数を計算
	return reduce!( ( a, b ) => a * b % MOD )( 1L, ms );
}
